

Lagrange Points


1 Discussion

No diagrams will be found in this work.

Joseph-Louis Lagrange, in the preface of his *Mécanique analytique* published 1788

While Joseph-Louis Lagrange (1736–1813) was proud to dispose of the geometrical language which Newton had used to write his *Principia*, we will discuss Lagrange’s mechanics using diagrams. Many of Lagrange’s efforts were directed at the *Three Body Problem*. While the motion of two point masses interacting through their mutual gravitational attraction is no more complicated than the motion of a single point-mass in a fixed $\frac{1}{r^2}$ force-field, the motion of three bodies interacting through gravity is a singularly difficult problem. Even today solutions take the form of approximations, rather than analytical results. Newton in working on the prototypical three-body problem: the motion of Sun, Earth, and Moon, said that it “made his head ache”. (We will mostly phrase our work in terms of three bodies: Earth, Moon, and satellite.)

We begin by considering the motion of the two most massive bodies (for us the Earth and Moon) and then move on to consider the motion of an object so light that it does not affect its partners. (Notice that this is a violation of Newton’s Third Law!) We will allow Newton’s Third to play a role for our two big bodies. Thus Moon pulls on the Earth exactly as strongly as the the Earth pulls on the Moon. . . the Earth is not an unmoved mover; it is also in orbit. The fixed point in the interaction between Earth an Moon is the center of mass.



We assume circular motion with (shared) angular speed ω , so:

$$M_1 r_1 = M_2 r_2$$

$$\begin{aligned} M_1 \omega^2 r_1 &= \frac{GM_1 M_2}{(r_1 + r_2)^2} \\ \omega^2 r_1 &= \frac{GM_2}{(r_1 + r_2)^2} \end{aligned}$$

$$\begin{aligned} M_2 \omega^2 r_2 &= \frac{GM_1 M_2}{(r_1 + r_2)^2} \\ \omega^2 r_2 &= \frac{GM_1}{(r_1 + r_2)^2} \end{aligned}$$

$$\omega^2 (r_1 + r_2) = \frac{G(M_1 + M_2)}{(r_1 + r_2)^2}$$

$$\omega^2 = \frac{G(M_1 + M_2)}{(r_1 + r_2)^3}$$

We now consider the view from a frame of reference rotating about the CM at exactly the pace needed to keep up with the Moon. In this frame, both Earth and Moon are at rest, but there are addition pseudo forces: centrifugal and Coriolis¹. A satellite moving in our new frame will follow:

$$m\mathbf{a} = -\frac{GM_1m}{|\mathbf{r} - \mathbf{r}_1|^2}\mathbf{u}_1 - \frac{GM_2m}{|\mathbf{r} - \mathbf{r}_2|^2}\mathbf{u}_2 + m\omega^2\mathbf{r} + 2m\omega\mathbf{v} \times \hat{\mathbf{z}}$$

where \mathbf{u}_1 and \mathbf{u}_2 are unit vectors that point towards Earth and Moon (respectively). Notice that the Coriolis force is just like the $\mathbf{v} \times \mathbf{B}$ magnetic force, the outward centrifugal force is like a sign-reversed spring force, and all forces are proportional to m . We now switch to units scaled to the problem. If T is the period of the Moon's orbit (i.e., $\omega = 2\pi/T$), then our dimensionless version of time, t' , is defined by:

$$t = t'T$$

Similarly we define a dimensionless distance:

$$\mathbf{r} = \mathbf{r}'(r_1 + r_2) = \mathbf{r}'d$$

If we switch to these variables our satellite's acceleration is given by:

$$\begin{aligned} \frac{d}{T^2} \mathbf{a}' &= \frac{G(M_1 + M_2)}{d^2} \left(-\frac{M_1/(M_1 + M_2)}{|\mathbf{r}' - \mathbf{r}'_1|^2} \mathbf{u}_1 - \frac{M_2/(M_1 + M_2)}{|\mathbf{r}' - \mathbf{r}'_2|^2} \mathbf{u}_2 \right) + \omega^2 d \left(\mathbf{r}' + 2\frac{1}{T\omega} \mathbf{v}' \times \hat{\mathbf{z}} \right) \\ &= \omega^2 d \left(-\frac{1 - \epsilon}{|\mathbf{r}' - \mathbf{r}'_1|^2} \mathbf{u}_1 - \frac{\epsilon}{|\mathbf{r}' - \mathbf{r}'_2|^2} \mathbf{u}_2 + \mathbf{r}' + \frac{1}{\pi} \mathbf{v}' \times \hat{\mathbf{z}} \right) \\ \mathbf{a}' &= (2\pi)^2 \left(-\frac{1 - \epsilon}{|\mathbf{r}' - \mathbf{r}'_1|^2} \mathbf{u}_1 - \frac{\epsilon}{|\mathbf{r}' - \mathbf{r}'_2|^2} \mathbf{u}_2 + \mathbf{r}' + \frac{1}{\pi} \mathbf{v}' \times \hat{\mathbf{z}} \right) \end{aligned}$$

where $\epsilon = M_2/(M_1 + M_2)$. (For the Earth-Moon system: $\epsilon = .0121$.) Except for the velocity-dependent Coriolis force we can make a potential to describe force ($-\nabla\phi = \mathbf{a}'$):

$$\phi(\mathbf{r}') = (2\pi)^2 \left(-\frac{1 - \epsilon}{|\mathbf{r}' - \mathbf{r}'_1|} - \frac{\epsilon}{|\mathbf{r}' - \mathbf{r}'_2|} - \frac{1}{2} r'^2 \right)$$


I'm tired of putting primes on everything; in what follows \mathbf{r} stands for \mathbf{r}' , etc.

```
ClearAll[e]
phi[x_,y_]=(2 Pi)^2 ( -(1-e)/Sqrt[(x+e)^2+y^2] -e/Sqrt[(x-(1-e))^2+y^2] -
(1/2)(x^2+y^2) )
fx[x_,y_]=-D[phi[x,y],x]
fy[x_,y_]=-D[phi[x,y],y]

e=.1
ContourPlot[phi[x,y],{x,-1.5,1.5},{y,-1.2,1.2},
Contours->10,PlotPoints->50,PlotRange->{-80,-50},AspectRatio->Automatic]
```

The contour plot shows a view similar to that looking down the cone of a double-sourced volcano. Well away from the origin, the centrifugal potential pushes everything out—that way is down the sides of the volcano cone. There are two “holes” which might source lava in a volcanic eruption—that way is down the gravitational well either to M_1 or M_2 . Around the edge of the cone is the crater lip. The lowest exit out is along the positive x axis (the connection between M_1 and M_2 is a bit lower), the exit along the negative x axis is the highest valley between the twin peaks at $\pm 60^\circ$ from M_1 . This shows this volcano:

¹France: Gustav-Gaspard Coriolis, 1835



```
Plot3D[phi[x,y],{x,-1.5,1.5},{y,-1.5,1.5},
PlotPoints->50,PlotRange->{-80,-50},ViewPoint->{2,-2,.6}]
```

Now to have a “stationary” satellite (in this rotating frame!) we must be at a point where $\mathbf{F} = \mathbf{0}$. Maximums, minimums, and “saddle points” provide $\nabla\phi = \mathbf{0}$. The saddle points don’t look like stable equilibrium points; the peaks look even less promising... but don’t forget that “magnetic field”. You should know that physicists (and Scotty on Star Trek) use magnetic fields to confine particles... Maybe it will stabilize our satellite!

But first let’s find the exact location of the peaks:

```
xp=-e+Cos[Pi/3]
yp=Sin[Pi/3]
Simplify[fx[xp,yp]]
Simplify[fy[xp,yp]]
```

The point indeed has $\mathbf{F} = \mathbf{0}$. From the origin, the point is back ϵ (i.e., on top of M_1), and then exactly 1 unit 60° from the x -axis. This proves that M_1 , M_2 , and our peak make an equilateral triangle. You might have hoped *Mathematica* could directly find the roots with a command like:

```
Solve[{fx[x,y]==0,fy[x,y]==0},{x,y}]
```

but the problem is too complex for *Mathematica* to find the roots without some help. In an appendix we will demonstrate the peaks are stable equilibrium positions for $\epsilon < 0.0385$. Instead let’s view satellite motion

```
e=.01
solution=NDSolve[{x'[t]==fx[x[t],y[t]]+(4 Pi) y'[t], y''[t]==fy[x[t],y[t]]-(4 Pi) x'[t],
x[0]==-e+Cos[Pi/3]+.01, y[0]==Sin[Pi/3],
x'[0]==0, y'[0]==0}, {x,y},{t,0,10}]
```

```
Out[11]= {{x -> InterpolatingFunction[{{0., 10.}}, <>],
> y -> InterpolatingFunction[{{0., 10.}}, <>]}}
```


```
ParametricPlot[Evaluate[{x[t],y[t]} /. solution],{t,0,10}]
ContourPlot[phi[x,y],{x,0,1.3},{y,0,1.3},ContourShading -> False,
Contours->20,PlotPoints->50,PlotRange->{-65,-59},AspectRatio->Automatic]
Show[%%,%,AspectRatio->Automatic]
```

```
e=.04
```

```
solution=NDSolve[{x''[t]==fx[x[t],y[t]]+(4 Pi) y'[t], y''[t]==fy[x[t],y[t]]-(4 Pi) x'[t],
  x[0]==-e+Cos[Pi/3]+.001, y[0]==Sin[Pi/3],
  x'[0]==0, y'[0]==0}, {x,y},{t,0,10}]
```

```
Out[17]= {{x -> InterpolatingFunction[{{0., 10.}}, <>],
  > y -> InterpolatingFunction[{{0., 10.}}, <>]}}
```

```
ParametricPlot[Evaluate[{x[t],y[t]} /. solution],{t,0,10}]
ContourPlot[phi[x,y],{x,0,1.3},{y,0,1.3},ContourShading -> False,
Contours->20,PlotPoints->50,PlotRange->{-65,-59},AspectRatio->Automatic]
Show[%%,%,AspectRatio->Automatic]
```



Thus with $\epsilon = .04$ and initial condition very close to the point where $\mathbf{F} = \mathbf{0}$ (above right) the motion seems to be diverging, whereas with $\epsilon = .01$ and initial condition further from the $\mathbf{F} = \mathbf{0}$ location (above left) the motion seems confined.

2 Homework

Complete on of the below options. Remember to turn in a printout showing each step as *Mathematica* solves the problem, in addition to any requested plots. For a review of basic orbit parameters see:

<http://www.physics.csbsju.edu/orbit/orbit.2d.html>

Option A: Moon Shot

On the day before Christmas, 1968, the Apollo 8 astronauts became the first humans to orbit a celestial body other than the Earth (one of very few ‘firsts’ for NASA that often seemed to be behind the Soviets). Apollo 8 actually fired retrorockets (of course near periselene) to begin the first of ten lunar orbits, but contingency plans for retro failure allowed for an unpowered figure 8 Earth return. Here you will plan such a trajectory. Starting from $\mathbf{r} = (.05, 0)$ (unrealistic, but it makes it easier), find the velocity needed to project the spacecraft around the Moon and back in the general direction of the Earth. Fuel is limited, so limit your starting speed to a maximum of 35. Determine the round trip time. Convert initial position, velocity, and round trip time into miles, mph, and days. While the rotating frame makes calculation easy, it’s interesting to view the trajectory in the Earth-relative frame, by shifting the origin to Earth and unrotating.

```
rot[t_]={{Cos[2 Pi t], -Sin[2 Pi t]},{Sin[2 Pi t],Cos[2 Pi t]}}
ParametricPlot[Evaluate[rot[t].{x[t]+e,y[t]} /. solution],{t,0,.3},AspectRatio->Automatic]
```

Provide nice hardcopies of your Moon Shot trajectories!

Option B: Lunar Slingshot & Energy

All numerical computation involves slight round-off error as only a finite number of digits can be calculated and retained. We know that energy is conserved, but in the numerical solutions energy is not exactly conserved and if the nonconservation is ‘significant’ the solution should be discarded. Work out a “Moon Shot” orbit and plot out energy as a function of time:

```
phi[x[t],y[t]]+(1/2)(x'[t]^2+y'[t]^2) /. First[solution]
Plot[%,{t,0,1}]
```

You should find nearly constant energy. Fast changing motions—as during periapsis (a generic term that includes close approaches to any body)—are hard on numerical differential equation solvers, and hence you’ll see glitches then.

Lets rethink that too quick claim of conservation of energy; what about gravitational sling shots (where spacecraft energy is increase with undetectable changes in planetary energy)? If we calculate actual energy (no centrifugal potential) in the inertial frame, what will result?

```
phi2[x_,y_]=(2 Pi)^2 ( -(1-e)/Sqrt[(x+e)^2+y^2] -e/Sqrt[(x-(1-e))^2+y^2] )
vx[t_]=x'[t]-y[t]2 Pi /. First[solution]
vy[t_]=y'[t]+x[t]2 Pi /. First[solution]
phi2[x[t],y[t]]+(vx[t]^2+vy[t]^2)/2 /. First[solution]
Plot[%,{t,0,1}, PlotRange->All]
```

For the Moon shot solution, you will typically clear signs of energy change. By (slight) adjustment of the Moon shot parameters, find a trajectory that does not return to Earth, and the spacecraft energy switches dramatically from negative to positive during a close Lunar approach, indicating a never-return result. Provide a nice hardcopy of your lunar slingshot trajectory, and the corresponding plot of showing actual energy going positive.

Option C: Moon Perturbations I

Without the Moon, an Earth satellite orbits the Earth in a fixed ellipse. To what extent does the Moon’s pull change this result? Plan on following an orbit for ten years (120 months). Starting from $\mathbf{r} = (.4, 0)$, view an Earth orbit in the inertial frame. You must back-rotate to reach the inertial frame (see Option A). Provide nice hardcopies showing an initial orbit, an orbit after 5 years, and an orbit after ten years.

```
e=.0121
rot[t_]={{Cos[2 Pi t], -Sin[2 Pi t]},{Sin[2 Pi t],Cos[2 Pi t]}}
v0=3; theta=Pi/2
solution=NDSolve[{x'[t]==fx[x[t],y[t]]+(4 Pi) y'[t],
y'[t]==fy[x[t],y[t]]-(4 Pi) x'[t],
x[0]==.4,y[0]==0,x'[0]==v0 Cos[theta],y'[0]==v0 Sin[theta]},
{x,y},{t,0,120},MaxSteps-> 500000,AccuracyGoal -> 12]
```

```
ParametricPlot[Evaluate[rot[t].{x[t]+e,y[t]} /. solution],{t,0,.15},AspectRatio->Automatic]
ParametricPlot[Evaluate[rot[t].{x[t]+e,y[t]} /. solution],{t,60,60.15},AspectRatio->Automatic]
ParametricPlot[Evaluate[rot[t].{x[t]+e,y[t]} /. solution],{t,119.85,120},AspectRatio->Automatic]
```

Using a ruler and pencil, carefully draw the major axis for each ellipse. Using a protractor, measure the angle of each major axis. Estimate the *precession rate* (changing orientation of the orbit: unit= degrees per year). Using Kepler’s Third Law (in the nice units: months and Earth-Moon distance) calculate the orbit period. Again using ruler measurements, see if you can detect changes in eccentricity ($e = \sqrt{1 - b^2/a^2}$) or major axis.

Option D: Moon Perturbations II

We ought to be able do better than paper and pencil in measuring the precessing orbit described in Option C above. At any instant the orientation and eccentricity of the orbit can be calculated from the Lenz vector²:

$$\mathbf{A} = \frac{(\mathbf{r} \times \mathbf{v}) \times \mathbf{v}}{(2\pi)^2} + \frac{\mathbf{r}}{|\mathbf{r}|}$$

In this formula \mathbf{r} and \mathbf{v} are measured in an inertial frame with the Earth at the origin, so we must displace our CM-centered coordinate \mathbf{r} to center on the Earth, add the velocity from rotation to the velocity measured in the rotating frame, and “unrotate” the vectors.

```
vx[t_]=x'[t]-y[t]2 Pi /. First[solution]
vy[t_]=y'[t]+x[t]2 Pi /. First[solution]
xe[t_]=x[t]+e /. First[solution]

lenz=rot[t].({vy[t](y[t]vx[t]-xe[t]vy[t]),vx[t](xe[t]vy[t]-y[t]vx[t])}/(2 Pi)^2 +
{xe[t],y[t]}/Sqrt[xe[t]^2+y[t]^2}) /. First[solution]
```

Explain the first three lines of this code. The magnitude of \mathbf{A} is the eccentricity; the direction of \mathbf{A} is apogee. By measuring on a hardcopy plot of an orbit, estimate the eccentricity ($e = \sqrt{1 - (b/a)^2}$). By plotting the magnitude of \mathbf{A} you can see the eccentricity directly; By plotting the angle of \mathbf{A} you can see the precession directly.

```
Plot[Norm[lenz],{t,0,120}]
Plot[ArcTan[lenz[[1]],lenz[[2]]],{t,0,120}]
```

In these plots you should see ‘fat lines:’ the instantaneous eccentricity and orientation are influenced by the instantaneous location of the Moon, but an average ‘secular’ trend should be evident.

Question: is it possible for perturbations from other planets, acting over millions of years, to increase the Earth’s eccentricity enough that the oceans would boil at perihelion? Lagrange’s answer in the work cited above, was “Non”.

Option E: Tides

A fluid Earth would conform to an isopotential surface. (Which isopotential depends on how much water volume exists.) In general the effect is small so we must “un-wrap” the Earth before we plot (N.B. radius of Earth=.0166; show this!):

```
ContourPlot[phi[(r+.0166)Cos[u]-e,(r+.0166)Sin[u]],{u,-Pi,Pi},{r,0,.00000001}]
```

The above plots the isopotential lines for a total r range of about 10 feet above a sphere centered on the Earth. Note that the isopotential stretches out about a foot at $u = \pm\pi, 0 \dots$ A fluid Earth would be extended towards the Moon ($u = 0$) and in the direction opposite the Moon ($u = \pm\pi$). Explain why this means we should have about two high tides per day. Unit convert to show that the size of the tides is approximately as I stated above.

Option F: Binary Stars & Mass Transfer

Consider a binary star system where $M_1 = 2M_2$, i.e., $\epsilon = \frac{1}{3}$. It turns out that, if these stars were born at the same time, M_1 will expand to become a red giant first. If the stars are separated by less than the distance M_1 seeks to expand into, some of the gas that makes up M_1 will be deposited onto M_2 . As stated above, fluid (e.g., gaseous) objects will try to conform to isopotentials. Thus as M_1 starts to expand, the low potential point between the stars will serve as the lip pouring material onto M_2 (i.e., surplus material for M_1 will start from the Lagrange point L_1 on a trip towards M_2).

²Lenz refers to W. Lenz who used this vector in his 1924 paper on the H-atom. However, Laplace’s 1799 work *Traité de mécanique celeste* contains a much earlier use of this “vector”

```
e=1/3
ContourPlot[phi[x,y],{x,-.6,1.1},{y,-.6,+.6},ContourShading->False,
Contours->20,PlotPoints->50,AspectRatio->Automatic]
L1=FindRoot[fx[x,0]==0,{x,.25}]
```

In the above plot you can see this L_1 point at about $x = .25$. `FindRoot` allows us to find the x location between Earth and Moon where $F_x = 0$; that is the unstable L_1 location (a saddle point). (Note: numerical root finding requires an initial guess. As above Mathematica's `Solve` can not use algebra to find this root.) Additionally note that the isopotentials are not exactly circles: a near-filling giant star would be elongated along the line connecting the stars due to the tidal force.

Perform a numerical solution to the differential equation starting material a bit beyond L_1 with zero velocity. Plot (and turn in) the motion of the material over a month `ParametricPlot`. What is the smallest M_2 radius that would result in the material just barely hitting "on the first try"? How long would it take to hit? (Think about plotting distance to M_2 vs. time.) To get another perspective, plot the 'unrotated' version of this motion (from release to hit) along with the trajectory of the 'Moon' during this same interval. (See Option A for unrotating.)

Option G: Mean Motion Resonance

When satellites have periods that are whole number fractions of a Moon's period, perturbations can have quite significant effects. In the rings of Saturn, we find gaps (no particles); for example, the Cassini Division is a gap that has been cleared by a 2:1 resonance with the moon Mimas. Three moons of Jupiter are forced to maintain a 1:2:4 orbital period ratio.

Here we look at the future of a Earth satellite with period $\frac{1}{2}$ of the Moon's. To examine the orbit over many cycles, we use the Lenz vector method of Option D.

```
e=.0121; x0=.621; v0=4
solution=NDSolve[{x'[t]==fx[x[t],y[t]]+(4 Pi) y'[t],
y'[t]==fy[x[t],y[t]]-(4 Pi) x'[t],
x[0]==x0, y[0]==0,
x'[0]==0, y'[0]==v0},
{x,y},{t,0,120},MaxSteps->50000,AccuracyGoal->12]

ParametricPlot[Evaluate[{x[t],y[t]}/.solution],{t,0,120},AspectRatio->Automatic]

vx[t_]=x'[t]-y[t]2 Pi /. First[solution]
vy[t_]=y'[t]+x[t]2 Pi /. First[solution]
xe[t_]=x[t]+e /. First[solution]

lenz=rot[t].({vy[t](y[t]vx[t]-xe[t]vy[t]),vx[t](xe[t]vy[t]-y[t]vx[t])}/(2 Pi)^2 +
{xe[t],y[t]}/Sqrt[xe[t]^2+y[t]^2}) /. First[solution]

ParametricPlot[lenz,{t,0,30}]
Plot[Norm[lenz],{t,0,120}]
Plot[ArcTan[lenz[[1]],lenz[[2]]],{t,0,120}]
```

Describe these results!

3 Appendix

We seek here to demonstrate the stability of the L_4 Lagrange point. We begin by Taylor expanding the potential in the vicinity of L_4 . Clearly near L_4 the potential must have the form:

$$\phi = \phi_0 + \frac{1}{2}Ax^2 + Bxy + \frac{1}{2}Cy^2$$

ϕ_0 has no effect on the forces, linear terms in the Taylor expansion are zero since by definition $\nabla\phi = \mathbf{0}$ there, and we neglect the higher order terms which will be quite small a few km (i.e., $\Delta x, \Delta y \sim 10^{-6}$) from L_4 . It turns out that we will have to work some to extract the homogeneous quadratic piece of the Taylor polynomial. If you remember your high school analytical geometry you should remember that by a suitable rotation of axes, the above form can be transformed to:

$$\phi = \phi_0 + \frac{1}{2}A'x'^2 + \frac{1}{2}C'y'^2$$

ClearAll[e]

Series[phi[x-e+Cos[Pi/3],y+Sin[Pi/3]],{x,0,2},{y,0,2}]

Normal[%] /. {x->q x, y->q y} ...so the terms we want will be q^2

Coefficient[%, q^2] ...OK we've got our quadratic form; now rotate axes

The easiest way to rotate our axes is to express the quadratic form in matrix form:

$$\frac{1}{2}Ax^2 + Bxy + \frac{1}{2}Cy^2 = \frac{1}{2} \begin{pmatrix} x & y \end{pmatrix} \begin{pmatrix} A & B \\ B & C \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} x & y \end{pmatrix} M \begin{pmatrix} x \\ y \end{pmatrix}$$

and then diagonalize M (i.e., find it's eigenvalues). The orthogonal matrix that diagonalizes M also rotates the coordinate system.

```
m={{Coefficient[%, x^2],Coefficient[%, x y]/2},{ Coefficient[%, x y]/2, Coefficient[%, y^2]}}
Eigensystem[%]
mdia=DiagonalMatrix[First[%]]
phiApprox[x_,y_]={y,x}.mdia.{y,x}
fxApprox[x_,y_]=-D[phiApprox[x,y],x]
fyApprox[x_,y_]=-D[phiApprox[x,y],y]
```

```
DSolve[{x''[t]==fxApprox[x[t],y[t]]+(4 Pi) y'[t], y''[t]==fyApprox[x[t],y[t]]-(4 Pi) x'[t]},
{x,y},t]
```

Mathematica's solution goes on for some time, but you will notice the following terms:

```
      2          2  2
Sqrt[2] Sqrt[-Pi  + Sqrt[1 - 27 e + 27 e ] Pi ] #1
E
```

```
      2          2  2
I Sqrt[2] Sqrt[Pi  + Sqrt[1 - 27 e + 27 e ] Pi ] #1
E
```

Clearly the nature of the solution changes if

$$1 - 27\epsilon + 27\epsilon^2 < 0$$

as it will be if $0.0385 \approx (9 - \sqrt{69})/18 < \epsilon < (9 + \sqrt{69})/18 \approx .9615$

Note there is nothing particularly difficult in solving this approximate differential equation:

$$\ddot{\mathbf{r}} = \begin{pmatrix} A' & 0 \\ 0 & C' \end{pmatrix} \cdot \mathbf{r} + 4\pi \begin{pmatrix} 0 & 1 \\ -1 & 0 \end{pmatrix} \cdot \dot{\mathbf{r}}$$

It can be solved by assuming a solution of the form: $\mathbf{r} = \mathbf{r}_0 e^{\gamma t}$, and then solving the “eigenproblem”:

$$\begin{pmatrix} A' - \gamma^2 & 4\pi\gamma \\ -4\pi\gamma & C' - \gamma^2 \end{pmatrix} \cdot \mathbf{r}_0 = 0$$